
AMS 394 one sample test (inference on one population mean)

1. SAS

- Open SAS: Now type in the following program in the SAS program editor. Note that comments can be included in the SAS program in the following fashion: /* comments */ (I always include more than one * to signify the comments).

```
/****** The Data Step *****/
```

```
data HTWT;  
input subject gender $ height weight;  
diff=height-68;  
datalines;  
1 M 68 155  
2 F 61 99  
3 F 63 115  
4 M 70 205  
5 M 68 170  
6 F 65 125  
7 M 72 220  
8 F 66 150  
;  
run;
```

```
/******The Proc Means Procedure******/
```

```
proc means data=HTWT;  
title 'Simple descriptive statistics';  
run;
```

```
proc means data=HTWT;  
var height;  
title 'Simple descriptive statistics for height';  
run;
```

```
proc means data=HTWT t prt;  
var diff;  
title 'One sample t-test to test whether the average height is 68 inches';  
title2 'that is, whether diff equals to 0';  
run;
```

```
proc means data=HTWT n mean stderr t prt maxdec=2;
var diff;
title 'One sample t-test with more options';
run;
```

```
/******The Proc TTEST Procedure******/
```

```
PROC TTEST DATA= HTWT;
VAR diff;
RUN;
```

```
/******The Proc Univariate Procedure******/
```

```
proc univariate data=HTWT;
var diff;
title 'More descriptive statistics and one-sample tests for diff';
run;
```

```
proc univariate data=HTWT normal plot;
var diff;
title 'including the normality tests and plots for diff';
run;
```

- Now run your SAS program and examine the output; check the SAS log if the program is not running properly.
- Save your SAS program and SAS output to a folder (directory) called “Lab”. File → Save as (for example, **htwt**). You can also save by clicking on the diskette like icon in the tool bar. Make sure the cursor is in the window that you want to save (that is, the window to be saved must be the active window). SAS will add the extension .sas for SAS program and .lst for SAS output.

Example 2:

The seven scores listed below are axial loads (in pounds) for a random sample of 7 12-oz aluminum cans manufactured by ALUMCO. An axial load of a can is the maximum weight supported by its sides, and it must be greater than 165 pounds, because that is the maximum pressure applied when the top lid is pressed into place.

270, 273, 258, 204, 254, 228, 282

(1) As the quality control manager, please test the claim of the engineering supervisor that the average axial load is greater than 165 pounds. Use $\alpha = 0.05$.

What assumptions are needed for your test?

(2) Please write a R program to do part (a).

(3) Please write a SAS program to do part (a).

Sol)

$$(1) \begin{cases} H_0 : \mu = 165 \\ H_a : \mu > 165 \end{cases}$$

Assume the distribution is normal.

$$T_0 = \frac{\bar{X} - \mu_0}{S/\sqrt{n}} \stackrel{H_0}{\sim} t_{n-1}$$

At the significance level of $\alpha = 0.05$, we reject H_0 in favor of H_a if $T_0 \geq t_{n-1, \alpha}$

$$T_0 = \frac{87.7}{27.6/\sqrt{7}} \doteq 8.9 > 1.943 : \text{We reject } H_0$$

$$CI : \bar{X} \pm t_{n-1, \alpha/2} \frac{S}{\sqrt{n}}$$

(2) This is how to do it in R:

```
# create a data frame from scratch
pressure <- c(270, 273, 258, 204, 254, 228, 282)
newvar = pressure - 165
mydata2 <- data.frame(pressure, newvar)
mydata2
```

Performs the Shapiro-Wilk test of normality.

```
shapiro.test(pressure)
```

Performs the one-sample t-test.

```
# one sample t-test
```

```
t.test(y, mu=3) # Ho: mu=3
```

```
t.test(pressure, mu=165)
```

(3) This is how to do it in SAS:

```
data cans ;  
input pressure @@ ;  
newvar = pressure - 165;  
datalines ;  
270 273 258 204 254 228 282  
;  
run ;  
  
proc univariate data=cans normal ;  
var newvar ;  
run ;  
  
proc ttest data=cans sides=upper ;  
var newvar ;  
run ;
```

(***Please note that by adding the @@ signs at the end of the input step, we will be able to enter data for different subjects on the same row, without having to separate the data for each subject by row.)