

Counting your data with PROC FREQ

The basic form of PROC FREQ is

```
PROC FREQ;  
  TABLES variable-combinations;
```

To produce a one-way frequency table, just list the variable name. To produce a cross-tabulation, list the variables separated by an asterisk.

Example The proprietor of a local coffee shop, Cathy's Coffee Cup, keeps a record of all sales. For each drink sold, she records the type of coffee (cappuccino, espresso, kona, or iced coffee), and whether the customer walked in or came to the drive-up window. The following program reads the data and produces one-way and two-way frequencies.

```
DATA orders;  
INPUT Coffee $ Window $ @@;  
DATALINES;  
esp w cap d cap w kon w ice w kon d esp d kon w ice d esp d  
cap w esp d cap d Kon d . d kon w esp d cap w ice w kon w  
kon w kon w ice d esp d kon w esp d esp w kon w cap w kon w  
;  
* Print tables for Window and Window by Coffee;  
PROC FREQ DATA = orders;  
  TABLES Window Window * Coffee;  
RUN;
```

The second table is a two-way cross-tabulation of Window by Coffee. Inside each cell, SAS prints the **frequency**, **percentage**, **percentage for that row**, and **percentage for that column**; while cumulative frequencies and percents appear along the right side and bottom. Notice that the missing value is mentioned but not included in the statistics.

```
DATA QUEST;  
INPUT ID 1-3 AGE 4-5 GENDER $ 6 RACE $ 7 MARITAL $ 8 EDUC $ 9  
PRES 10 ARMS 11 CITIES  
12; DATALINES;  
001091111232  
002452222422  
003351324442  
004271111121  
005682132333  
006651243425  
;  
PROC FREQ DATA=QUEST;  
  TITLE 'FREQUENCY COUNTS FOR CATEGORICAL VARIABLE';  
  TABLES GENDER RACE MARITAL EDUC PRES ARMS CITIES;  
RUN;
```

*Two-way frequency tables ;

```
PROC FREQ DATA=QUEST;  
  TABLES EDUC GENDER EDUC*GENDER;  
RUN;
```

Computing Chi-square from Frequency Counts

When you already have a contingency table and want to use SAS software to compute a chi-square statistic, there is a WEIGHT statement that makes this task possible

```
DATA CHISQ;  
    INPUT GROUP $ OUTCOME $ COUNT;  
DATALINES;  
CONTROL DEAD 20  
CONTROL ALIVE 80  
DRUG DEAD 10  
DRUG ALIVE 90  
;  
PROC FREQ DATA=CHISQ;  
    TABLES GROUP*OUTCOME / CHISQ;  
    WEIGHT COUNT;  
RUN;
```